
Úvod 

6 přednáška Algortimy

Toto je pouze pracovní materiál. Není zde uveden kompletní obsah přednášky.

Euklidův algoritmus

Známý algoritmus na určení největšího společného dělitele NSD dvou čísel A a B. Menší číslo odečítám od většího tak dlouho dokud obě nejsou rovná.

Musím několikrát opakovaně provádět stejnou činnost – použiji cyklus.

Do … Loop cyklus

Cykly slouží k opakovanému provádění některých částí programu. Obecným cyklem je v jazyce Visual Basic cyklus Do ... Loop. Tento cyklus lze použít v několika variantách podle toho jaká podmínka cyklus ukončuje.

Příkad: Cyklus s podmínkou na začátku provádění cyklu:

Do While podmínka


příkazy

Loop

Příkazy se v tomto cyklu opakují tak dlouho, dokud není splněna podmínka.

Podobně lze vložit podmínku na konec cyklu Do ... Loop While podmínka.

Někdy nemůžeme podmínku umístit ani na začátek, ani na konec cyklu. Pak je možné použít právě obecný cyklus Do ... Loop s tím, že cyklus ukončíme kdekoliv uvnitř cyklu pomocí příkazu Exit Do. Samozřejmě tento příkaz je vhodné umístit do podmíněného příkazu. 
Dim A As Integer, B As Integer
Dim NSD As Integer 'nejvetsi spolecny delitel cisel A a B

Do While A <> B
    If A > B Then
        A = A - B
    Else
        B = B - A
    End If
Loop
NSD = A

Prvočíslo

Algoritmus, který určí zda zadané celé číslo je prvočíslo. Je dělitelné pouze 1 a samo sebou. Číslo zkouším opakovaně dělit čísly (2, ... ?) a zkoumám zbytky po celočíselném dělení. Pokud je zbytek 0, číslo není prvočíslo.

Číslo získám od uživatele programu pomocí prvku InputBox.

Funkce InputBox

Tato funkce zobrazí dialogové okno s textovým řetězcem, který obsahuje instrukci pro uživatele ohledně hodnoty, kterou má zadat. Kromě tohoto řetězce je zde i textové pole, do kterého uživatel zapíše požadovanou hodnotu. 

Ukázka volání funkce InputBox

Form1.Caption = InputBox("Zadej název formuláře:", "Název")

Uživatel zadá řetězec, který bude dosazen do záhlaví formuláře.

InputBox(prompt[, title] [, default],[další parametry])

prompt – text, který se zobrazí uživateli

title – titulek okna

default – předdefinovaná hodnota, nabídnutá uživateli

o dalších parametrech si můžete přečíst v nápovědě

Pokud uživatel stiskne tlačítko Cancel, dosadí se do výsledku prázdná hodnota. Další zpracování musíme ošetřovat programem.

Používáme ho všude tam, kde od uživatele potřebujeme získat nějaké informace.

Ukázka programu

Dim Cislo As Integer 'zkoumane cislo
Dim I As Integer    'delitele
Dim prvocislo As Boolean
Cislo = Val(InputBox("Zadej číslo:", "Prvočíslo", 13)) 'uzivatel zada cislo, nastaveno 13
prvocislo = True
I = 2
Do While I <= Sqr(Cislo) And prvocislo
    If (Cislo Mod I) = 0 Then prvocislo = False
    I = I + 1
Loop
If prvocislo Then
    MsgBox "Číslo " & Cislo & " je prvočíslo"
Else
    MsgBox "Číslo " & Cislo & " není prvočíslo"
End If

Číselná řada (rady/rady1.vbp)

Uživatel zadá řadu kladných celých čísel zakončenou číslem-1, spočtěte jejich součet.

První nad čím je potřeba se zamyslet je jakým způsobem umožním uživateli zadat řadu čísel. Protože nevíme kolik čísel je, nemůžeme si pomocí nástrojů, které máme připravit vstupní prvky. Otázkou také je, zda by se nám tyto prvky vešly na formulář.

Můžeme prvky zadávat buď pomocí InputBoxu nebo je můžeme mít zapsané v textovém souboru a načítat tyto hodnoty z textového souboru. Na přednášce si ukážeme obě varianty.

Private Sub Command1_Click()
Dim A As Integer 'číslo
Dim Max As Integer, Min As Integer
Dim S As Integer
Dim i As Integer 'pomocná proměnná
i = 1
A = Val(InputBox("Zadej čís." & i & vbNewLine & 

"Zadávání ukonči číslem -1", "Zadávej celá čísla"))
Do Until A = -1

S = S + A

i = i + 1

A = Val(InputBox("Zadej čís." & i & vbNewLine & 

"Zadávání ukonči číslem -1", "Zadávej celá čísla"))
Loop

If i > 0 Then

Command1.Visible = False

lbsoucet.Caption = S

lbsoucet1.Visible = True

lbsoucet.Visible = True
End If

End Sub

Jak načítat čísla ze souboru. 

Musíme se naučit pracovat s textovým souborem.

Pokud chceme k datům v souboru přistupovat sekvenčně (zpracujeme je postupně od začátku do konce) můžeme použít .

Open "data.txt" For Input As #1 'otevreni souboru data.txt

'v kodu soubor vystupuje jako cislo 1

Z tohoto souboru data čteme pomocí příkazu Input

Input #1, A

‘načtení hodnoty za souboru #1 do proměnné A

Tento příkaz soubor uzavře:

Close #1

Celý kód načtení a sečtení hodnot ze souboru pak vypadá takto:

Private Sub Command1_Click()
Dim A As Integer 'číslo
Dim S As Integer
S = 0

Open "data.txt" For Input As #1 'otevreni souboru data.txt
'v kodu soubor vystupuje jako cislo 1

Do While Not (EOF(1)) 'dokud neni konec souboru
    Input #1, A       'precte jedno cislo
    S = S + A
Loop

Close #1
Command1.Visible = False
lbsoucet.Caption = S
lbsoucet1.Visible = True
lbsoucet.Visible = True


End Sub































